

Paper 254-2009

Designing Multi-Step Fractional Factorial Split-Plots: A Combined JMP® and SAS® User Application

José G. Ramírez, Ph.D., W.L. Gore and Associates Inc., Landenberg, PA

Jon T. Weisz, SAS Institute Inc., Cary, NC

ABSTRACT

In the past 10 years, there has been an increase in research and software development in the design of experiments for split-plot situations. As supply chains grow across the globe and complexity increases, it is necessary to design experiments for processes involving three, four, or more steps. New features in PROC FACTEX in SAS® 9.2 make it possible to design fractional factorial split-plots for multi-step situations. This paper gives a brief introduction on how to design fractional factorial split-plots for multi-step situations using real industrial examples, and shows how JMP® users can take advantage of these new capabilities using a JMP® application.

INTRODUCTION

Complex manufacturing situations not only impose constraints in the number of experiments that can be run, but also make it very difficult, or impossible, to fully randomize all the experimental runs as required by good experimental design practices. In addition, multiple sizes of experimental units can result as raw materials and intermediate products pass from one manufacturing step to the next. Designing experiments in this environment requires us to take into account the number of steps, along with the constraints on experimental materials, time, etc. Fortunately for us recent developments in JMP and SAS make it easier to generate experimental designs for these situations.

MULTI-STEP PROCESSES AND SPLIT-PLOTS DESIGNS

A multi-step process is one in which different manufacturing steps are carried in sequence either in the same location, but more often, in different plants, or even across the world. At each process step different process parameters can be changed as shown in Figure 1




Figure 1. Multi-Step Process

One quickly realizes that the number of experimental factors can become quite large, and that a complete, or full factorial, experiment is out of the question. In some cases, in order to manage this complexity, experiments are designed for each process step but this usually leads to sub-optimization of the complete multi-step system. What is needed is a holistic approach that takes into account each process step and its group of factors, and the interactions between all the factors in the multi-step process. Understanding these interactions is critical for the global optimization of the multi-step process.

Consider a simple 2-step process situation in which lots of different Film Types (F1 or F2) are manufactured at a given Plant A. Rolls from each lot are then sent to Plant B for further processing under three different Pressures (P8, P11, P12). An experiment to understand the effects of Film Type, Pressure and their interaction will require six different combinations of the 2 Film Types and the 3 Pressures. Figure 2 shows the treatment structure for the Film Type x Pressure experiment




Figure 2. Treatment Structure for Film Type x Pressure Experiment Showing 6 Lots and 18 Rolls

For the experiment, six lots, three of Film Type F1 and three of Film Type F2, were selected at random at Plant A, and each lot was “split” into 3 rolls that were processed at Plant B using one of the three Pressures: P8, P11, or P12. Note that there are two different sizes of experimental units (EUs), the Lot manufactured at Plant A using the factor Film Type, and the Roll processed at Plant B using the factor Pressure. A full randomization is not possible because we first have to randomize the lots coming from Plant A, and *then* randomize which rolls, from a given lot, will be processed using one of the 3 pressures. The different sizes of experimental units induce a restriction in randomization leading to a split-plot experimental structure. Figure 3, shows the ANOVA table for this experiment.

SOURCE	DF		
Film Type	1	2-1	
Lot(Film Type)	4	$2 \times (3-1)$	3 Lots per Film Type
Pressure	2	3-1	
Film Type x Pressure	2	$(2-1) \times (3-1)$	
Roll(Film Type x Pressure)	8	$2 \times 3 \times (3-1)$	3 Rolls per Film Type x Pressure Combination
Total	17	18-1	

Figure 3. ANOVA Table for Film Type x Pressure Experiment

It is very important to realize that this design have two error terms, reflecting the variation between Lots (EU1), and the variation between Rolls within Lots (EU2). The ANOVA table is “split” into two sections corresponding to the two steps, or the 2 experimental units. The top part corresponds to the 6 lots manufactured in Plant A, with a total of 5 degrees-of-freedom, while the bottom part corresponds to the 18 rolls processed with the 6 Film Type x Pressure combinations.

In general, split-plot designs occur primarily in two situations (see Ramírez (2004)):

1. There is a *restriction in randomization* for an experimental factor.
For example, running all treatment combinations associated with “low” temperatures followed by all treatment combinations associated with “high” temperatures.
2. There are *different sizes of experimental units*.
For example, in the example above Film Type, was applied to a Lot (experimental unit 1); while Pressure was applied to a roll coming from the lot (experimental unit 2). The Lot was “split” into three Rolls for the experimental factor Pressure.

As you can see, multi-step processes give rise to split-plot type structures. The larger the number of process steps, the larger the number of “splits” in the experimental structure.

DESIGNING EXPERIMENTS FOR MULTI-STEP PROCESSES

Restrictions in randomization or different sizes of experimental units, along with the constraints on the number of resources available make designing experiments for multi-step processes a challenge. We need ways of designing experiments that take into account the split-plot structure induced by multi-step processes, as well as the requirement of trying to use a reasonable number of runs. Two approaches are possible: we can use fractional factorial designs, or we can specify a model along with an optimality criteria to obtain a design. Currently with versions 7 and 8, JMP users can take advantage of the optimality criteria approach to design experiments for 2-Step and 3-Step processes. But what if you process has more than 3 steps? Or what if you want a design for a 2-

Step or 3-Step situations with different characteristics than the ones offered by JMP? Recent advances in fractional factorial split-plots (FFSP) methods and software make it now possible to design experiments for virtually any number of process steps.

FRACTIONAL FACTORIAL SPLIT-PLOTS (FFS^KP)

Fractional factorial designs are a very efficient way to run experiments subject to constraints on the number of observations. The idea of constructing fractional factorials to take into account the split-plot confounding is not new. Kempthorne (1952) called these types of designs fractional factorial split-plot, and Addelman (1964) provided a table of factorial plans with split-plot confounding for different number of Whole-Plot and Split-Plot factors. However, in recent years there has been a wave of research on fractional factorial split-plots (FFSP). Huang et al (1998), and Bingham and Sitter (1999, 2001) have applied the concept of *minimum aberration* (Fries and Hunter (1980)) to split-plot designs, giving comprehensive tables for small to moderately sized minimum aberration split-plot designs, while Kulahci et al (2006) have discussed alternative criteria to minimum aberration for designing FFSP.

Useful catalogs of FFSP have appeared in the pages of various journals. For example, Huang, Chen, and Voelkel (1998) give a catalog of minimum aberration fractional split-plot designs. In this era of great computing power we do not need to be limited by looking at tables for a design, we can let the computer find an appropriate design given design and model constraints. In SAS® Version 9.2 PROC FACTEX enables us to construct custom FFSP designs for situations with *multi-step* processes, or with more complex restrictions in randomization, and multiple sizes of experimental units. The intuitive syntax allows multiple specifications of wholeplots and subplots, making it possible to cover a wide array of situations (See also, Ramirez and Tobias (2007)).

A 3-STEP PROCESS EXAMPLE

Consider a situation where a process is run in three steps. In Step 1 there is one factor that we can change, Z. In Step 2 the material coming from Step 1 is processed according to seven process factors (A, B, C, D, E, F, G). Finally, in Step 3 the product reaches its final form by means of 8 process factors (P, Q, R, S, T, U, V, W). This 3-step process gives rise to a split-split-plot structure.

JMP CUSTOM DESIGN

The “Custom Design” (DOE > Custom Design) in JMP 8 can be used to design a split-split-plot type experiment to estimate all the main effects, and all the two factor interactions. Figure 4 below shows the inputs to the “Custom Designer”

The screenshot shows the 'DOE - Custom Design' dialog box. It is divided into several sections:

- Custom Design:** Includes 'Responses' and 'Factors' sections. The 'Factors' section has a table with columns: Name, Role, Changes, and Values.

Name	Role	Changes	Values
Z	Categorical	Very Hard	L1, L2
A	Categorical	Hard	L1, L2
B	Categorical	Hard	L1, L2
C	Categorical	Hard	L1, L2
D	Categorical	Hard	L1, L2
- Define Factor Constraints:** Includes a 'Model' section with buttons for 'Main Effects', 'Interactions', 'RSM', 'Cross', 'Powers', and 'Remove Term'. Below this is a table for 'Estimability':

Name	Estimability
Intercept	Necessary
Z	Necessary
A	Necessary
B	Necessary
C	Necessary
D	Necessary
E	Necessary
F	Necessary
- Design Generation:** Includes a checkbox for 'Hard to change factors can vary independently of Very Hard to change factors.' Below this are input fields for 'Number of Whole Plots' (8) and 'Number of Subplots' (38).
- Number of Runs:** Includes radio buttons for 'Minimum' (137), 'Default' (256), and 'User Specified' (256).

Figure 4. Custom Design Inputs for $2^1 \times 2^7 \times 2^8$ Experiment with 8 Whole Plots and 38 Subplots.

By default, JMP generates a 256-run design in which all the two factor interactions are clear of confounding with main effects and other two factor interactions. This experiment has 8 Whole Plots or 8 runs of the Step-1 factor Z, and 38 Subplots, or 38 runs of the Step-2 factors A, B, C, D, E, F, G.

PROC FACTEX

If JMP was able to generate a D-optimal design for the 3-Step process described above, why do we need to use fractional factorials? The answer depends on the characteristics of the JMP generated design, and how the experiment is going to be run. The (default) 256-run design generated by JMP's Custom Designer is unbalanced. The 8 whole plots do not have the same size: there are 6 whole plots of size 5, and 2 whole plots of size 4. For the 38 subplots we have 6 subplots of size 4, 24 subplots of size 7, and 8 subplots of size 8. Also, although all the main effects and 2-factor interactions are clear from confounding with other 2-factor interactions, the main effects in this D-optimal design are confounded with (long) strings of 3-factor interactions. Given that we have 16 factors in this experiment, and due to the complexity of the manufacturing process, it is quite possible for some of the 560 3-factor interactions to have an effect. Therefore, in this situation, it is desirable to have a design where the main effects are at clear from 2-factor AND 3-factor interactions.

The number of whole plots and subplots dictate how the experiment needs to be conducted. The Step-1 factor has to be changed 7 times, there are 8 runs (whole plots) of this factor; while for the 7 factors in Step-2 we need to run 38 different combinations, the 38 subplots. The structure needs to be taken into account to make sure that is feasible, in terms of time and resources, to run the experiment that way. Note also that JMP's 8 Custom Designer took over 35 minutes in an Intel® Core™ Duo CPU E6850 3GHz; 3GHz to generate this design.

Fractional factorial designs are an alternative to optimal designs. We can take advantage of the new features in PROC FACTEX, SAS/QC version 9.2, view to generate a 256-run fractional factorial split-split-plot for the 3-Step process. The new features consist of the BLOCK UNIT=() option for describing the split-plot structure of the experiment, and the UNITEFFECT statement for specifying where effects of interest should be estimable within this scheme. We denote a 3-Step fractional factorial split-plot as FFS²P. The superscript 2 denotes the number of process steps after the first step, or the number of "splits", hence the split-split-plot. The PROC FACTEX syntax is as follows:

```

/* Factors for the 21x27x28 Experiment */
%let F1 = Z;
%let F2 = A B C D E F G;
%let F3 = P Q R S T U V W;
proc factex ranorder(restart seed=1);
  factors &F1 &F2 &F3;
  /* Resolution 4, minimum aberration */
  model r=4 / minabs;
  /* This defines the FFS2P: 4 x 21x27-4x28-6 Experiment */
  /* Structure has 8 Whole Plots and 8 Subplots */
  blocks units=(Step1=8 Step2=8);
  /* Step 1 */
  uniteffect Step1          / whole=(&F1) sub=(&F2 &F3);
  /* Step 2 */
  uniteffect Step1*Step2 / whole=(&F2) sub=(      &F3);
  /* 256 runs - Only 0.4% of the total 65536 runs*/
  size design=256;
  examine confounding aliasing abs;
run;
quit;

```

The resulting design has 256 runs, but the structure is different from the optimal design generated by JMP. This is a balanced design with 8 Whole Plots, each of size 8, and 64 (8*8) Subplots, each of size 4. In terms of execution, the Step-1 factor also has to be changed 7 times, as in the D-optimal design, and for the 7 factors in Step-2 we need to run 8 experiments for each of the 8 runs of the Step-1 factor Z.

Apart from being balanced the main advantage of this FFS²P design is that is a resolution 5 design; i.e., ALL the 16 main effects are clear from confounding with any 2-factor AND 3-factor interaction, and ALL 120 2-factor interactions are clear from confounding with main effects and other 2-factor interactions. PROC FACTEX took about 1 minute in an Intel® Core™ Duo CPU E6850 3GHz; 3GHz to find this design.

You may have noticed that the PROC FACTEX FFS²P design has 64 subplots, while the default JMP design has 38. In order to make a more apples-to-apples comparison we can overwrite the default choices and ask the JMP Custom Designer to generate a design with 8 Whole Plots and 64 Subplots as shown in Figure 5

Figure 5. Custom Design Inputs for $2^1 \times 2^7 \times 2^8$ Experiment with 8 Whole Plots and 64 Subplots.

The resulting JMP design is balanced and has the same structure of the FACTEX one; i.e., 8 whole plots of size 8 each, and 64 subplots of size 4 each. However, all the main effects of this design are still confounded with long strings of 3-factor interactions. The generation of this design in JMP took over 50 minutes in an Intel® Core™ Duo CPU E6850 3GHz; 3GHz.

DESIGNING MULTI-STEP PROCESSES EXPERIMENTS USING JMP-SAS INTEGRATION

With JMP version 7 a new era of integration between JMP and SAS was made possible. JMP has been able to read and write SAS data sets for many versions, but JMP was not able to connect or submit SAS code to PC SAS on the same machine, or connect and submit SAS code to a remote server. This meant that JMP users could not leverage the deep data management or statistical features available in other SAS products.

At the same time, SAS could not take advantage of the JMP Scripting Language's (JSL) ability to create dynamic user interfaces or JMP's interactive graphics.

By developing integration between JMP and SAS allowing "point and click" users of JMP to query and retrieve data from a SAS server as well as execute SAS reports (SAS Stored Processes), new access to corporate data assets and canned reports were available to JMP users for the first time.

The integration also offers the SAS programmer the ability to package SAS code within JSL and deploy to JMP users within the organization. This means that SAS programmers leveraging the statistical power of SAS/QC and the user experience of the JMP environment can add new features, like support for Fractional Factorial Split-Plot designs, to JMP.

We have developed JSL code for designing fractional factorial experiments for multi-step processes situations, to take advantage of the new features in PROC FACTEX in SAS/QC using a native JMP interface. These capabilities enhance the Custom Design capabilities available in JMP 8, giving the user more design choices.

Below we give 3 examples of multi-step processes situations to show the applicability of the techniques, as well as the ease of use on the JMP-SAS application.

A 3-STEP PROCESS EXAMPLE

Consider the multi-step process situation in the previous section where a process is run in three steps. In Step 1 there is one process factor that we can change, Z. In Step 2 the material coming from Step 1 is processed according to seven process factors (A, B, C, D, E, F, G). Finally, while in Step 3 the product reaches its final form by manipulating 8 process factors (P, Q, R, S, T, U, V, W). We want to design an experiment in 256 runs (only 0.4% of the 65,536 runs for the full factorial), with 8 runs of the Step 1 factor (Z), 8 runs of the Step 2 factors (A, B, C, D, E, F, G) for each run of factor Z, and 4 runs of the Step 3 factors (, Q, R, S, T, U, V, W) for each Step 1 and Step 2 run.

Figure 6 shows the JMP interface for designing FFS²P experiments. In this window we input the process factors for each of the 3 steps, as well as the structure of the experiment in terms of the

1. Number of Runs for Step 1 Factors
2. Number of Runs for Step 2 Factors Within Step 1 Factors
3. Total Number of Runs




Figure 6. JMP Interface for Designing 3-Step Process Experiments.

When we click “OK” a connection to SAS is established, and the information in Figure 6 is sent to PROC FACTEX in order to search for a design with those characteristics. The output consists of the design generators, the alias structure for the design, the word length pattern, and a JMP table with the FFS²P design. A nice option in the JMP-SAS integration is the ability to convert SAS output into a familiar JMP report. The (partial) alias structure in Figure 7 shows all the main effects clear from confounding with 2-factor and 3-factor interactions, and the 2-factor interactions clear from confounding with other two factor interactions; in other words, a resolution 5 design.




Figure 7. PROC FACTEX Output for FFS²P Experiment.

Figure 8 shows the JMP with the fractional factorial split-split-plot design. There are two columns, Step1 and Step 2 that index, from 1 through 8, the number of runs for the Step1 factors, and the number of runs for Step2 factors

	Step1	Z	Step2	A	B	C	D	E	F	G	P	Q	R	S
1	1	-1	1	1	-1	-1	-1	1	1	-1	1	-1	-1	1
2	1	-1	1	1	-1	-1	-1	1	1	-1	-1	1	1	1
3	1	-1	1	1	-1	-1	-1	1	1	-1	-1	1	1	-1
4	1	-1	1	1	-1	-1	-1	1	1	-1	-1	-1	-1	-1
5	1	-1	2	-1	-1	-1	1	-1	-1	1	1	-1	1	-1
6	1	-1	2	-1	-1	-1	1	-1	-1	1	-1	1	-1	-1
7	1	-1	2	-1	-1	-1	1	-1	-1	1	1	1	1	-1
8	1	-1	2	-1	-1	-1	1	-1	-1	1	-1	-1	1	1
9	1	-1	3	1	-1	-1	-1	-1	1	1	-1	-1	-1	1
10	1	-1	3	1	-1	-1	-1	-1	1	1	1	1	1	1
11	1	-1	3	1	-1	-1	-1	-1	1	1	-1	1	1	-1
12	1	-1	3	1	-1	-1	-1	-1	1	1	1	-1	-1	-1
13	1	-1	4	-1	-1	-1	1	1	-1	-1	-1	-1	1	-1
14	1	-1	4	-1	-1	-1	1	1	-1	-1	1	1	1	-1
15	1	-1	4	-1	-1	-1	1	1	-1	-1	-1	1	-1	1
16	1	-1	4	-1	-1	-1	1	1	-1	-1	1	-1	1	1
17	1	-1	5	1	1	1	1	1	-1	-1	1	-1	1	1
18	1	-1	5	1	1	1	1	1	-1	-1	-1	1	-1	1
19	1	-1	5	1	1	1	1	1	-1	-1	1	1	-1	-1
20	1	-1	5	1	1	1	1	1	-1	-1	-1	-1	1	-1
21	1	-1	6	-1	1	1	-1	-1	1	1	1	-1	-1	-1
22	1	-1	6	-1	1	1	-1	-1	1	1	-1	1	1	-1
23	1	-1	6	-1	1	1	-1	-1	1	1	1	1	1	1
24	1	-1	6	-1	1	1	-1	-1	1	1	-1	-1	-1	1
25	1	-1	7	1	1	1	1	-1	-1	1	-1	-1	1	1
26	1	-1	7	1	1	1	1	-1	-1	1	1	1	-1	1
27	1	-1	7	1	1	1	1	-1	-1	1	-1	-1	-1	-1
28	1	-1	7	1	1	1	1	-1	-1	1	1	-1	1	-1
29	1	-1	8	-1	1	1	-1	1	-1	-1	-1	-1	-1	-1

Figure 8. JMP Table with Design for 3-Step Process Example.

4-STEP PROCESS EXAMPLE

The product manufactured in this 4-Step example starts with the raw material at Step 1, where it gets processed using 7 factors: A B C D E F G. After Step 1, the output goes through 3 passes of second process step. In the first pass 4 process factors are applied (Step 2): Q1 R1 S1 T1; in the second pass 5 factors (Step 3): Q2 R2 S2 T2 U; and in the third pass 4 factors (Step 4): Q3 R3 S3 T3. With 20 factors, the total number of runs for a full factorial experiment is impossible ($2^{20} = 1,048,576$). A design to estimate all the 20 main effects clear from confounding with other 2-factor interactions, and as many 2-factor interactions clear from other 2-factor interactions was needed. Figure 9 shows the input window for the JMP-SAS application for generating a fractional factorial designs for this 4-Step process situation.

4-Step Process (FF53P) Design Generator

PROCESS FACTORS

Step 1 Factors: A B C D E F G

Step 2 Factors: Q1 R1 S1 T1

Step 3 Factors: Q2 R2 S2 T2 U

Step 4 Factors: Q3 R3 S3 T3

EXPERIMENTAL STRUCTURE

Number of Runs for Step 1 Factors: 16

Number of Runs for Step 2 Factors Within Step 1 Factors: 2

Number of Runs for Step 3 Factors Within Step 1xStep 2 Factors: 2

Total Number of Runs: 128

Buttons: Generate Design, OK, Cancel, Remove, Help

Figure 9. JMP Interface for Designing 4-Step Process Experiments.

The inputs Figure 9 indicate that we want to generate a 128-run design with 16 unique runs for the 7 Step 1 factors, and 2 runs each for the Step 2 and Step 3 factors, within Step 1 and Step 2 respectively. Figure 10 shows the JMP table with the design.

	Step1	A	B	C	D	E	F	G	Step2	Q1	R1	S1	T1	Step3
1	1	1	-1	-1	1	1	-1	-1	1	-1	-1	1	1	1
2	1	1	-1	-1	1	1	-1	-1	1	-1	-1	1	1	1
3	1	1	-1	-1	1	1	-1	-1	1	-1	-1	1	1	2
4	1	1	-1	-1	1	1	-1	-1	1	-1	-1	1	1	2
5	1	1	-1	-1	1	1	-1	-1	2	1	1	-1	-1	1
6	1	1	-1	-1	1	1	-1	-1	2	1	1	-1	-1	1
7	1	1	-1	-1	1	1	-1	-1	2	1	1	-1	-1	2
8	1	1	-1	-1	1	1	-1	-1	2	1	1	-1	-1	2
9	2	1	-1	1	-1	-1	1	-1	1	-1	1	1	-1	1
10	2	1	-1	1	-1	-1	1	-1	1	-1	1	1	-1	1
11	2	1	-1	1	-1	-1	1	-1	1	-1	1	1	-1	2
12	2	1	-1	1	-1	-1	1	-1	1	-1	1	1	-1	2
13	2	1	-1	1	-1	-1	1	-1	2	1	-1	-1	1	1
14	2	1	-1	1	-1	-1	1	-1	2	1	-1	-1	1	1
15	2	1	-1	1	-1	-1	1	-1	2	1	-1	-1	1	2
16	2	1	-1	1	-1	-1	1	-1	2	1	-1	-1	1	2
17	3	-1	1	-1	-1	1	1	-1	1	-1	1	-1	1	1
18	3	-1	1	-1	-1	1	1	-1	1	-1	1	-1	1	1
19	3	-1	1	-1	-1	1	1	-1	1	-1	1	-1	1	2
20	3	-1	1	-1	-1	1	1	-1	1	-1	1	-1	1	2
21	3	-1	1	-1	-1	1	1	-1	2	1	-1	1	-1	1
22	3	-1	1	-1	-1	1	1	-1	2	1	-1	1	-1	1
23	3	-1	1	-1	-1	1	1	-1	2	1	-1	1	-1	2
24	3	-1	1	-1	-1	1	1	-1	2	1	-1	1	-1	2
25	4	-1	1	1	1	-1	-1	-1	1	-1	-1	-1	-1	1
26	4	-1	1	1	1	-1	-1	-1	1	-1	-1	-1	-1	1
27	4	-1	1	1	1	-1	-1	-1	1	-1	-1	-1	-1	2
28	4	-1	1	1	1	-1	-1	-1	1	-1	-1	-1	-1	2
29	4	-1	1	1	1	-1	-1	-1	2	1	-1	-1	-1	1

Figure 10. JMP Table with Design for 4-Step Process Example.

The above design is only 0.01% of the total number of runs for the full factorial. This design has all the 20 main effects clear from confounding with 2-factor interactions, and allows us to estimate 104 out of the total 190 2-factor interactions, 70 of them clear from confounding with other 2-factor interactions.

5-STEP PROCESS EXAMPLE

As a final example consider a 5-Step process with 15 factors distributed as follows:

- Step-1 Factors: Z
- Step-2 Factors: A B C D E F G
- Step-3 Factors: Q1 T1
- Step-4 Factors: Q2 T2 U
- Step-5 Factors: Q3 T3

We want to design a fractional factorial experiment in 128 runs that takes into account the 5 steps in the process. Figure 11 shows the required inputs for the JMP-SAS application.




Figure 11. JMP Interface for Designing 5-Step Process Experiments.

The corresponding design is shown in Figure 12. This design has all 15 main effects clear from confounding with 2-factor interactions, and allows us to estimate.

	Step1	Z	Step2	A	B	C	D	E	F	G	Step3	Q1	T1	Step4	Q2
1	1	-1	1	1	1	1	-1	-1	-1	1	1	-1	-1	1	1
2	1	-1	1	1	1	1	-1	-1	-1	1	1	-1	-1	1	1
3	1	-1	1	1	1	1	-1	-1	-1	1	1	-1	-1	2	-1
4	1	-1	1	1	1	1	-1	-1	-1	1	1	-1	-1	2	-1
5	1	-1	1	1	1	1	-1	-1	-1	1	2	1	1	1	1
6	1	-1	1	1	1	1	-1	-1	-1	1	2	1	1	1	1
7	1	-1	1	1	1	1	-1	-1	-1	1	2	1	1	2	-1
8	1	-1	1	1	1	1	-1	-1	-1	1	2	1	1	2	-1
9	1	-1	2	1	-1	1	-1	1	1	-1	1	1	1	1	-1
10	1	-1	2	1	-1	1	-1	1	1	-1	1	1	1	1	-1
11	1	-1	2	1	-1	1	-1	1	1	-1	1	1	1	2	1
12	1	-1	2	1	-1	1	-1	1	1	-1	1	1	1	2	1
13	1	-1	2	1	-1	1	-1	1	1	-1	2	-1	-1	1	-1
14	1	-1	2	1	-1	1	-1	1	1	-1	2	-1	-1	1	-1
15	1	-1	2	1	-1	1	-1	1	1	-1	2	-1	-1	2	1
16	1	-1	2	1	-1	1	-1	1	1	-1	2	-1	-1	2	1
17	1	-1	3	-1	-1	-1	-1	-1	-1	-1	1	1	1	1	1
18	1	-1	3	-1	-1	-1	-1	-1	-1	-1	1	1	1	1	1
19	1	-1	3	-1	-1	-1	-1	-1	-1	-1	1	1	1	2	-1
20	1	-1	3	-1	-1	-1	-1	-1	-1	-1	1	1	1	2	-1
21	1	-1	3	-1	-1	-1	-1	-1	-1	-1	2	-1	-1	1	1
22	1	-1	3	-1	-1	-1	-1	-1	-1	-1	2	-1	-1	1	1
23	1	-1	3	-1	-1	-1	-1	-1	-1	-1	2	-1	-1	2	-1
24	1	-1	3	-1	-1	-1	-1	-1	-1	-1	2	-1	-1	2	-1
25	1	-1	4	-1	1	-1	-1	1	1	1	1	-1	-1	1	-1
26	1	-1	4	-1	1	-1	-1	1	1	1	1	-1	-1	1	-1
27	1	-1	4	-1	1	-1	-1	1	1	1	1	-1	-1	2	1
28	1	-1	4	-1	1	-1	-1	1	1	1	1	-1	-1	2	1

Figure 12. JMP Table with Design for 5-Step Process Example.

FACTEX CODE

For those readers familiar with SAS, below is the PROC FACTEX code needed to generate the fractional factorial designs for the three examples presented above.

3-STEP PROCESS EXAMPLE

```

/* Factors for the 21x27x28 Experiment */
%let F1 = Z;
%let F2 = A B C D E F G;
%let F3 = P Q R S T U V W;
proc factex ranorder(restart seed=1);
  factors &F1 &F2 &F3;
  /* Resolution 4, minimum aberration */
  model r=4 / minabs;
  /* This defines the FFS2P: 4 x 21x27-4x28-6 Experiment */
  /* Structure has 8 Whole Plots and 8 Subplots */
  blocks units=(Step1=8 Step2=8);
  /* Step 1 */
  uniteffect Step1 / whole=(&F1) sub=(&F2 &F3);
  /* Step 2 */
  uniteffect Step1*Step2 / whole=(&F2) sub=( &F3);
  /* 256 runs - Only 0.4% of the total 65536 runs*/
  size design=256;
  examine confounding aliasing abs;
run;
quit;

```

4-STEP PROCESS EXAMPLE

```

/* Factors for the 27x24x25x24 Experiment */
%let F1 = A B C D E F G;
%let F2 = Q1 R1 S1 T1;
%let F3 = Q2 R2 S2 T2 U;
%let F4 = Q3 R3 S3 T3;
proc factex ranorder(restart seed=1);
  factors &F1 &F2 &F3 &F4;
/* Resolution 4, minimum aberration */
model r=4 / minabs;
/* This defines the FFS3P: 27-3x24-3x25-4x24-3 Experiment */
blocks units=(Step1=16 Step2=2 Step3=2);
/* Step 1 */
uniteffect Step1 / whole=(&F1) sub=(&F2 &F3 &F4);
/* Step 2 */
uniteffect Step1*Step2 / whole=(&F2) sub=( &F3 &F4);
/* Step 3 */
uniteffect Step1*Step2*Step3 / whole=(&F3) sub=( &F4);
/* 128 runs - Only 0.01% of the total 1,048,576 runs*/
size design=128;
examine confounding aliasing abs;
run;
quit;

```

5-STEP PROCESS EXAMPLE

```

/* Factors for the 21x27x22x23x22 Experiment */
%let F1 = Z;
%let F2 = A B C D E F G;
%let F3 = Q1 T1;
%let F4 = Q2 T2 U;
%let F5 = Q3 T3;
proc factex ranorder(restart seed=1);
  factors &F1 &F2 &F3 &F4 &F5;
/* Resolution 4, minimum aberration */
model r=4 / minabs;
/* This defines the FFS3P: 21x27-4x22-1x23-1x22-1 Experiment */
blocks units=(Stage1=2 Stage2=8 Stage3=2 Stage4=2);
/* Step 1 */
uniteffect Stage1 / whole=(&F1) sub=(&F2 &F3 &F4 &F5);
/* Step 2 */
uniteffect Stage1*Stage2 / whole=(&F2) sub=( &F3 &F4 &F5);
/* Step 3 */
uniteffect Stage1*Stage2*Stage3 / whole=(&F3) sub=( &F4 &F5);
/* Step 4 */
uniteffect Stage1*Stage2*Stage3*Stage4 / whole=(&F4) sub=( &F5);
/* 128 runs - Only 0.4% of the total 32,768 runs*/
size design=128;
examine confounding aliasing abs;
run;
quit;

```

CONCLUSION

The JMP Custom Designer is a powerful tool that allows users to create a designed experiment to match their experimental situation. There are situations however, where a problem cannot be solved by JMP's powerful design tools, or the design offered by JMP can be improved upon to better match the experimental situation. The good news is that by integrating the design tools available in JMP and the design tools available in SAS/QC, users are now able to use the most powerful design of experiments workbench available.

REFERENCES

- Addelman, S. (1964). "Some Two-Level Factorial Plans with Split-Plot Confounding," *Technometrics* 6, 253-258.
- Bingham, D. and Sitter, R. R. (1999). "Minimum Aberration Two-level Fractional Factorial Split-Plot Designs". *Technometrics*, 41, 62-70.
- Bingham, D. and Sitter, R. R. (2001). "Design Issues in Fractional Factorial Split-Plot Designs". *Journal of Quality Technology*, 33, 2-15. John Wiley & Sons, Inc.
- Fries, A. and Hunter, W.G. (1980) "Minimum Aberration 2k-p Designs," *Technometrics*, 22, 601-608.
- Huang, P., Chen, D.; and Voelkel, J. (1998). "Minimum Aberration Two-level Split-Plot Designs". *Technometrics* 40, 314-326.
- Kempthorne, O. (1952). *The Design and Analysis of Experiments*. New York: John Wiley & Sons, Inc.
- Kulahci, M, Ramírez, J.G., and Tobias, R. (2006) "Split-Plot Fractional Designs: Is Minimum Aberration Enough?," *Journal of Quality Technology*, 31, 56-64.
- Ramírez, J.G., (2004) "To Split or not to Split: Do We Have a Choice?," Joint Newsletter for the Section on Physical and Engineering Sciences and the Quality and Productivity Section of the American Statistical Association, 11, Number 2, 10-12.
- Ramírez, J.G., and Tobias, R. (2007) "Split and Conquer! Using SAS/QC® to Design Quality into Complex Manufacturing Processes," SAS Global Forum Paper 190-2007.

ACKNOWLEDGMENTS

We wish to thank Randy Tobias for his helpful discussions about PROC FACTEX and fractional factorial split-plot experiments.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the authors at:

Name: José G. Ramírez
 Enterprise: W.L. Gore and Associates, Inc.
 Address: 380 Starr Road
 City, State ZIP: Landenberg, PA, 19350
 Work Phone: 610 268-1805
 Fax:
 E-mail: jramirez@wlgore.com
 Web: www.gore.com

Name: Jon Weisz
 Enterprise: SAS Institute Inc
 Address: SAS Campus Dr.
 City, State ZIP: Cary, NC 27513
 Work Phone: 919 531 3576
 Fax: 919 677 4444
 E-mail: jon.weisz@jmp.com
 Web: www.jmp.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.